
ABSTRACT  

The Add Health Parent Study is using a new and innovative method to augment our other interview verification 
strategies. Typical verification strategies include calling respondents to ask questions about their interview, recording 
pieces of interaction (CARI - Computer Aided Recorded Interview), and analyzing timing data to see that each 
interview was within a reasonable length. Geocoding adds another tool to the toolbox for verifications. By applying 
street-level geocoding to the address where an interview is reported to be conducted and comparing that to a 
captured latitude/longitude reading from a GPS tracking device, we are able to compute the distance between two 
points. If that distance is very small and time stamps are close to each other, then the evidence points to the field 
interviewer being present at the respondent’s address during the interview. For our project, the street-level geocoding 
to an address is done using SAS® PROC GEOCODE. Our paper describes how to obtain a US address database 
from the SAS website and how it can be used in PROC GEOCODE. We also briefly compare this technique to using 
the Google Map API and Python as an alternative. 

INTRODUCTION  

SAS Proc Geocode is being utilized on the Add Health Parent Study as a new tool to provide contributing 
information to managers of field studies who are responsible for ensuring that the field interviewers are 
conducted honestly.  Proc Geocode has provided a valuable addition to the set of tools used for interview 
verification and validation. 

RTI International has conducted hundreds of research studies involving field data collection on laptop 
computers.  Computer-Assisted Personal Interviewing (CAPI) is conducted by hiring a large staff of field 
interviewers to visit research participants in their homes.  These interviews range from the very short and 
simple to the very long and complex.  Sensitive questions are frequently asked. 

In spite of our best attempts to hire quality field interviewers, we sometimes find ourselves with an 
interviewer that will not follow the established protocols.  Some interviewers may even be tempted to 
cheat, especially on projects where each interview is long, complex, and emotionally demanding.  
Cheating can involve anything from leading respondents to answers they would not have otherwise given, 
all the way to complete interview falsification.  Nothing can be more damaging to a research project than 
fabrication of data.  At RTI, we take interview falsification very seriously and we have developed 
multipronged procedures to detect it quickly. 

The following methods are typically used to detect falsification. 

 CARI – Computer-Assisted Recorded Interviewing is a technique used to record portions of field 
interviews using the laptop microphone.  As audio files are returned to RTI, a research 
professional listens to each one in order to make sure the voice of a respondent is heard. 

 Whole Interview Timing Data – Interviews that are not legitimate will sometimes have outlier 
length values far from the average length of time the interview normally takes. 

 Specific Question Timing Data – Lengthy questions, such as questions about consent to the 
interview, should show evidence that the field interviewer spent enough time on the question for it 
to be covered adequately.  For example, spending 3 seconds on a consent question that includes 
several pages of text would be considered suspicious.   

 Verification Calls – When phone numbers are collected in the course of completing an interview, 
sometimes verification calls are made.  If the phone numbers are wrong or disconnected, this 
raises suspicion that the interview was falsified. 

The Add Health Parent Study has added interview geocoding to the set of tools used for detecting the 
possibility of interview falsification.  Each field interviewer is carrying a DG-100 geocoding device.  At the 

conclusion of each interview, field interviewers are instructed to capture the latitude and longitude of the 
interview location as they leave.  By using the location data captured and making a comparison with the 
geo-coordinates of the physical address using SAS version 9.4, we are able to determine if the field 
interviewer was actually present at the address for the interview.  Time stamps are also part of the data 
capture.  By comparing the time stamp of the geo-coordinates to the time stamp captured on the main 
interview, we can determine if they were close together as expected. 

It is important to understand that a discrepancy in geo-coded data is not, in itself, enough evidence to 
determine that there is anything wrong with the interview in question.  However, a discrepancy in the geo-
coded data can be detected very quickly and call attention to the case so that the other verification 
methods can be included in the analysis of the case’s legitimacy.  For example, the person reviewing 
CARI recordings can prioritize cases with questionable geo-coded data and listen to those CARI 
recordings first.  The report can also be easily modified to group all observations done by the same field 
interviewer in order to see if an outlier GPS reading is part of an emerging pattern or a true outlier. 

IMPLEMENTATION 

We selected the DG-100 for capture of geo-coordinates because of its simplicity and low cost.  It is also less intrusive 
than something like a geo-tagged photograph of the home.  We purchased 35 units at an average cost of $22 each.  
Each comes with a USB plug to download readings to any USB capable laptop or desktop.  There is no LED display.  
Field interviewers are simply instructed to turn it on, wait for a few minutes until the green light flashes, and then 
press the silver button. 

 

Display 1: DG-100 

 

After the coordinates have been captured, the field interviewers plug the DG-100 into their laptop and 
launch another interview module called the Field Interviewer Observations Module.  The GPS readings 
and associated data are downloaded into this interview module using free software called GPSBabel.  

Display 2: Screen Image of GPS readings downloaded into Blaise interactive interview software 

 

Information about GPSBabel can be found on website www.gpsbabel.org. During download two files 
associated with the interview are created on the laptop: .csv and .kml.  Excel can be used to review the 
.csv file and the .kml file can be viewed in Google Earth as seen below.  In this .kml file, the DG-100 was 
left running and the whole route captured. 

 

 

Display 3: .kml file viewed in Google Earth 

   

 

 

 

http://www.gpsbabel.org/

PROCESSING 

When getting started with Proc Geocode, we found several good examples on the SAS website.  The 
default database SAS/GRAPH uses for street level geocoding is SASHELP.GEOEXP and includes 
addresses from Wake County in North Carolina, where the SAS headquarters is located.  To get the data 
needed to geocode throughout the United States, SAS provides a free download from its SAS Maps 
Online website.   

http://support.sas.com/documentation/cdl/en/graphref/65389/HTML/default/viewer.htm#n02y3yabtlqatsn1
6gp2fo51yo7p.htm#n1nmfmy3wf1kc8n0zw565lb41mjt 

The complete street level lookup file is called USM.sas7bdat and comes with an associated index file.  
For information regarding geocoding addresses outside the US, there was a good paper presented at the 
2013 SAS Global Forum that you may reference. 

http://support.sas.com/rnd/papers/sasgf13/Geocode2013.pdf 

The data files available are version dependent.  The example that follows runs with SAS version 9.4 but 
will terminate with errors when run under earlier versions.  In this example, relevant data is pulled from 
two interview datasets.  Case assignment data is pulled from an SQL database that supports our 
integrated field management system (IFMS) so that each case can be associated with a particular field 
interviewer.  The two sets of geo-coordinates are used to calculate the distance between the two points 
using two methods.  Finally, the SAS Output Delivery System (ODS) is used with Proc Print to summarize 
the results.  Interviews showing the greatest distance between the two points are sorted to the top of the 
report.  These are the interviews that will receive the greatest scrutiny through the other methods of 
interview verification and validation. 

/* Code segments found online are highlighted below with indications about where they were found. */ 

/* must run under v 9.4 */ 

libname out1 '\\rtishare\AddHealthPS\RCD\SASPrgs\'; 

libname in2 '\\rtishare\AddHealthPS\RCD\DataMgt\SASoutput\fir\'; 

libname in1 'C:\Geocode\Data\'; 

libname library '\\rtishare\AddHealthPS\RCD\DataMgt\SASoutput\FIR\'; 

libname in3 '\\rtishare\AddHealthPS\RCD\DataMgt\SASoutput\AddhPS\'; 

libname field ODBC complete='DRIVER=SQL Server;SERVER=RTPWSRVER\PRIVPROD,1433; 
DATABASE=AHP; Trusted_Connection=True;' schema=fld; 

 

/* The interview files have many formats permanently associated with the answers to interview questions 
so nofmterr is used. */ 

options nofmterr; 

 

 

 

 

 

 

 

 

http://support.sas.com/documentation/cdl/en/graphref/65389/HTML/default/viewer.htm#n02y3yabtlqatsn16gp2fo51yo7p.htm
http://support.sas.com/documentation/cdl/en/graphref/65389/HTML/default/viewer.htm#n02y3yabtlqatsn16gp2fo51yo7p.htm
http://support.sas.com/rnd/papers/sasgf13/Geocode2013.pdf

/* Proc SQL is used to retrieve case assignment data so that interviews can be attributed to specific 
interviewers. */ 

proc SQL; 

create table assign as 

select a.Res_ID as caseid, a.ToSite as FIID,p1.Name as FIName from Field.Assignment a inner join 
Field.v_4_ProjectAssignment p1  

on a.ToSite=p1.FieldID 

where a.FldCode=2690 and a.RTI2Fistat>0 and a.Training=0 and p1.IsActive='Y' 

order by caseid; 

quit; 

 

/* Physical addresses are retrieved from the FI Observations interview module for geocoding. */ 

data myGPS(keep= Caseid address city state zip); 

   set in2.fir(rename=(zrid=caseid address1=address)); 

   if GPSData_Date = ' ' then delete; 

   zip = input(zipcode,5.); 

run;    

 

/* Proc Geocode is called with myGPS as the input dataset and out1.geocoded as the output 
dataset. */ 

/* This puts the Latitude and Longitude of the physical address in the output dataset. */ 

/* The following code was found on the SAS support website.  Example #3 Street Geocoding. */ 

proc geocode 

   method=STREET 

   data=myGPS 

   out=out1.geocoded2 

   lookupstreet=in1.usm 

/* lookupstreet=SASHELP.GEOEXM - this is the default value before downloading full reference 
files. */ 

   attributevar=(TRACT);    

run; 

 

proc sort data=out1.geocoded2; by caseid; run; 

 

/* This retrieves the GPS data from the FI Observations interview dataset so it can be compared to the 
location of the physical address. */ 

data FI_Obs(keep=caseid GPSdata_Latitude GPSdata_Longitude GPSdata_date GPSdata_time); 

   set in2.fir(rename=(zrid=caseid)); 

run; 

proc sort data=FI_Obs; by caseid; run; 

 

data out1.geodist2; 

   merge out1.geocoded2(in=aaa) FI_Obs(in=bbb) assign; 

   by caseid; 

   if aaa and bbb; 

    

   lat1 = y; 

   long1 = x; 

 

   lat2 = GPSdata_Latitude; 

   long2 = GPSdata_Longitude; 

 

   linkid = substr(caseid, 1, 7) || '1'; 

 

 /* The following distance calculation code came straight from sasCommunity.org */ 

/*www.sascommunity.org/wiki/Driving_Distances_and_Drive_Times_using_SAS_and_Google_Maps 

*/  

  * GEODIST function; 

   Dist = geodist(lat1, long1, lat2, long2, 'M'); 

    

   * convert latitude and longitude from degrees to radians for use in Haversine formula; 

 

   d_to_r = constant('pi')/180; 

   lat1  = lat1  * d_to_r; 

   long1 = long1 * d_to_r; 

   lat2  = lat2  * d_to_r; 

   long2 = long2 * d_to_r; 

    

   * HAVERSINE formula (earth radius of 3949.99 miles used to produce distance in miles); 

 

   Haversine = 3949.99 * arcos(sin(lat1) * sin(lat2) + cos(lat1) * cos(lat2) * cos(long2 - long1)); 

   put caseid "Distance to RTP using geodist function:" Dist; 

   put caseid "Distance to RTP using Haversine formula:" Haversine; 

 

run; 

data main_interview(keep=zrid Interview_compdate Interview_comptime); 

   set in3.addhps(rename=(comp_time=Interview_comptime)); 

   Interview_compdate = MDY(substr(comp_date,1,2), substr(comp_date,3,2), substr(comp_date,5,4)); 

run;  

 

data out1.geodist3; 

   merge main_interview(in=aaa) out1.geodist2(in=bbb rename=(linkid=zrid)); 

   by zrid; 

   if aaa and bbb;    

run; 

 

proc contents data=out1.geodist3; 

   title 'geodist'; 

run; 

 

proc sort data=out1.geodist3; 

   by descending Haversine; 

run; 

 

ods html file='\\rtishare\AddHealthPS\RCD\DataMgt\Reports\Rpt_GeoDistanceNew.xls' style=sasweb; 

 

proc print data=out1.geodist3 noobs split='*'; 

   var caseid address M_addr M_zip _matched_ _status_ _score_ lat1 long1 lat2 long2 dist haversine 
_notes_ Interview_compdate Interview_comptime GPSdata_date GPSdata_time ; 

   label M_addr = 'Matched*Address' 

         M_zip = 'Matched*Zip' 

         _matched_='Match*Method' 

         _status_='Match*Status' 

         _notes_='Notes' 

         _score_='Match*Score' 

         Dist='Calculated*Distance' 

         haversine='Haversine*Distance' 

         lat2='Captured Lat' 

         long2='Captured Long' 

         Interview_compdate='Interview*Comp Date' 

         Interview_comptime='Interview*Comp Time' 

         GPSdata_date='Capture*Date' 

         GPSdata_time='Capture*Time'; 

   title 'Distance between captured geo-coordinates and physical address'; 

   footnote 'Refer to the following link for a key to notes values'; 

   footnote2 '\\rtishare\AddHealthPS\RCD\Reports\Street Geocoding Note Values.docx'; 

   format Interview_compdate mmddyy10.;  

run; 

    

ods html close; 

 

OUTPUT 

The following report output is for illustrative purposes only and uses simulated data.  Latitude and 
longitude readings have been converted to radians for distance calculation. 

 

  Output 1: Left side of report 

Output 2: Right side of report 

 

The first thing to note about this output is that not every address can be found.  When the address is not 
found, the Proc Geocode reverts to returning the geo-coordinates of the “centroid” of the zip code.  This 
can be some distance from the actual address.  The “match score” of 15 is a low score and reflects the 
fact that the address could not be found.  These observations are generally the ones showing the 
greatest distance from the address, but knowing the reason helps the data collection manager 
understand the output and to not overreact to the reported distance between the two points.   

Looking at the observation listed first on the report, we can also note that the time stamps are not very 
close to each other (as seen in the right panel on the far right).  The programming built into the program 
that downloads and processes the GPS readings looks for the observation from the GPS unit that is 
closest to the current date and time.  In this case, there was no reading close to that.  This indicates that 
the field interviewer may not have initially been getting readings or forgot to take the reading.  By 
comparing this result to the other observations the particular field interviewer was responsible for, we 
would be able to determine if this is a pattern.  The interviewer could be contacted for their input, 
retraining could be needed, and possibly the replacement of faulty hardware could be needed. 

The code abbreviations found in the “codes” column of the report provide detailed information about the 
results of the address matching.  The codes used factor into the match score.  The SAS support website 
explains the relationship between the codes and score as follows. 

“The STREET geocoding method creates a _NOTES_ variable in the output data set. This variable 
provides details about the quality of the address match by using token strings. For example, the value 
"AD ZC NM" contains three tokens that indicate that the street name, ZIP code, and house number 
matched. 

Each token in the _NOTES_ value has an associated value, and the sum of these values makes up the 
value of the _SCORE_ variable.” 

The common abbreviations are shown below.  A complete list of these abbreviations and their meanings 
can be found on the SAS support website. 

Token Score Description 

AD 20  The street name matched. 

DP 15  The street direction prefix matched. 

ENDNM  0 The house number was outside the ranges of values in the lookup data 
set for the matching street. The geocoded coordinates for the nearest 
end of the street were used. 

NM 10  The house number matched on the correct side of the street. 

NOCTM  0 A match was found using the input address’ street and ZIP code.  
However, it was noted that the input city value was different from the city 
on the matched lookup dataset observation. 

NODPA -10 The input address had no direction prefix but the matched street did 
have a direction prefix. 

NODPM -15 The input address had no direction prefix but it either did not match the 
direction prefix of the matching street or the matching street had no 
direction prefix. 

NMOS  5 The house number matched an address range in the lookup data set, but 
is on the opposite side of the street from the matched range. 

TS 20  The street type prefix matched. 

ZC 15  The five-digit ZIP code or non-U.S. postal code matched. 

 

CONCLUSION 

This new method of providing additional information for detecting falsification has not led directly to 
finding falsification.  However, it has added information that has been in favor of confirming the legitimacy 
of cases. The true value of this processing as an interview validation method may not be in catching 
falsification but in helping to prevent its occurrence in the first place.  As field interviewers become aware 
that the legitimacy of their interviews will be scrutinized in multiple ways including the capture of their 
location and timing data, they should understand that cheating is not an option.  The end result should be 
to keep them on the straight and narrow path to a clean and honest set of interviews. 

APPENDIX 

Alternative geocoding implementation using Google Maps API 

In this section, we’ll briefly explore the Google Maps API as an alternative to SAS PROC GECODE. The 
Google Maps API provides multiple comprehensive services, including Mapping, Directions, Places and 
Geolocation. The APIs support desktop and mobile devices, and in all popular programming languages: 
Java, C#, JavaSript, Python, etc. Our code samples will be focused on the Geocoding service using 
Python. An excellent introduction to the Python language can be found on this website: 

https://www.python.org/about/gettingstarted/ 

The following code sample illustrates how to use the Google Maps API to obtain the following geolocation 
data from a given address: 

Standardized address 
Latitude/longitude coordinate 
Match type: partial, rooftop, range interpolation, geometric center 
 

from pygeocoder import Geocoder  

 

address = '303 Gingergate Dr Cary NC' 

results = Geocoder.geocode(address)  

 

print('Formatted address: {}'.format(results.formatted_address))  

print(‘Lat/Long: {}’.format(results.coordinates)) 

print(‘Location type: {}’.format(results.location_type)) 

 

Obtain geolocation data using Google Geolocation API 

 

This program produces the following output: 

Formatted address: 303 Gingergate Dr, Cary, NC 27519, USA 

Lat/Long: (35.7867109, -78.8531859) 

Location type: ROOFTOP 

 

 First, we import the Geocoder object from the pygeocoder module which is a Python library that wraps 
the Google's Geocoding functionality. 

 We then call the geocode method from the Geocoder object to obtain geolocation data. 

 Finally we print the geocoding results: formatted address, latitude/longitude coordinate, and location 
type. ROOFTOP indicates that the location information is accurate down to street address precision. 
Other location type’s values are: RANGE_INTERPOLATED, GEOMETRIC_CENTER, and 
APPROXIMATE.  

  

https://www.python.org/about/gettingstarted/

Reverse geocoding (Address Lookup) 

The Google Geolocation API also includes reverse geocoding functionality that can be used for looking 
up address from a geo-coordinate. This is illustrated in the following code sample. 

 

from pygeocoder import Geocoder 

 

lat, lng = (35.7867109, -78.8531859) 

results = Geocoder.reverse_geocode(lat, lng) 

print('Formatted address: {}'.format(results.formatted_address)) 

print('Location type: {}'.format(results.location_type)) 

 

The result of this program is: 

Formatted address: 303 Gingergate Dr, Cary, NC 27519, USA 

Location type: ROOFTOP 

 

 

As shown, reverse geocoding can be easily done with the reverse_geocode method and similar 

geolocation data can be obtained from this method. 

As illustrated, the Google Geolocation API is straightforward to use. In our evaluation, it produces results 
comparable to those obtained from the SAS PROC GEOCODE in most situations. A unique functionality 
supported by the Google API that is not available in the SAS procedure is the ability to do reverse 
geocoding. Further documentation about Google Maps API is available at this website: 

https://developers.google.com/maps/web-services/overview 

 

https://developers.google.com/maps/web-services/overview

